
vcgen/Basic.v

Set Implicit Arguments.
Require Export ZArith.
Require Export List.
Open Scope list_scope.

Definition value := Z.

(** Binary Operators *)

Inductive bop : Set :=
  | Oadd
  | Omul.

Definition eval_bop bop : value -> value -> value :=
  match bop with
  | Oadd => Zplus
  | Omul => Zmult
  end.

(** Comparison Operators *)
 
Inductive cop : Set :=
  | Olt
  | Ole
  | Oeq.

Definition eval_cmp cmp : value -> value -> bool :=
  match cmp with
  | Olt => Zlt_bool
  | Ole => Zle_bool
  | Oeq => Zeq_bool
  end.

Definition interp_cmp cmp : value -> value -> Prop := 
  match cmp with 
  | Olt => Zlt
  | Ole => Zle
  | Oeq => (@eq Z)
  end.

Lemma eval_cmp_true_interp : forall c v1 v2,
   eval_cmp c v1 v2 = true -> interp_cmp c v1 v2.
Proof.
 destruct c;simpl.
 intros v1 v2 H;assert (X:= Zlt_cases v1 v2);rewrite H in X;trivial.
 apply Zle_bool_imp_le.
 unfold Zeq_bool;intros.
 apply Zcompare_Eq_eq.
 destruct ((v1 ?= v2)%Z);trivial;try discriminate.
Qed.

Lemma eval_cmp_false_interp : forall c v1 v2,
   eval_cmp c v1 v2 = false -> ~ interp_cmp c v1 v2.
Proof.
 destruct c;simpl;intros v1 v2 H H0.
 assert (X:= Zlt_cases v1 v2);rewrite H in X;auto.
 rewrite Zle_imp_le_bool in H;[discriminate H|trivial].
 unfold Zeq_bool in H;rewrite H0 in H.
 rewrite Zcompare_refl in H;discriminate H.
Qed.

(** Logical operators *)

Inductive lop : Set :=
  | Limp
  | Land.

Definition interp_lop o :=
  match o with
  | Limp => fun (P1 P2: Prop) => P1 -> P2
  | Land => and
  end.

Section GENERIC.

 Variable V : Set.

 (** A generic type for expressions *)

 Inductive gexpr : Set :=
   | Const (z:Z)
   | Var (x:V)
   | Op (e1:gexpr) (o:bop) (e1:gexpr).


 (** A generic type for propositions *)

 Inductive prop : Set :=
   | Ltrue 
   | Lfalse
   | Lnot (p:prop)
   | Lcmp (e1:gexpr) (c:cop) (e2:gexpr)
   | Lop (p1:prop) (op:lop) (p2:prop).

 Section INTERP.
 
  Variable rho: V -> value.
 
  Fixpoint eval_gexpr (e:gexpr) : Z :=
    match e with
    | Const z => z
    | Var x => rho x
    | Op e1 o e2 => eval_bop o (eval_gexpr e1) (eval_gexpr e2)
    end.

 Fixpoint interp_prop (p:prop) : Prop :=
   match p with
   | Ltrue => True
   | Lfalse => False
   | Lnot p => not (interp_prop p)
   | Lcmp e1 c e2 => interp_cmp c (eval_gexpr e1) (eval_gexpr e2)
   | Lop p1 o p2 => interp_lop o (interp_prop p1) (interp_prop p2)
   end.

 End INTERP.

End GENERIC.

Section MAP.

 Variables V1 V2:Set.
 Variables f : V1 -> gexpr V2.

 Fixpoint map_gexpr (e:gexpr V1) : gexpr V2 :=
   match e with
   | Const z => Const _ z
   | Var x => f x 
   | Op e1 o e2 => Op (map_gexpr e1) o (map_gexpr e2)
   end.

 Fixpoint map_prop (P:prop V1) : prop V2 :=
   match P with  
   | Ltrue => Ltrue _ 
   | Lfalse => Lfalse _
   | Lnot P => Lnot (map_prop P)
   | Lcmp e1 c e2 => Lcmp (map_gexpr e1) c (map_gexpr e2)
   | Lop p1 o p2 => Lop (map_prop p1) o (map_prop p2)
   end.

End MAP.


(** Variables *)

Definition var := positive.

Definition Veq x y := 
  match Pcompare x y Eq with
  | Eq => true
  | _ => false
  end.

(** Variables appearing in postcondition *)

Inductive pvar : Set :=
 | PV0 (x:var)
 | PVres.
 
(** Memory operations *)

Definition mem := var -> value.

Definition update (m:mem) (x:var) (v:value) : mem:=
  fun y => if Veq x y then v else m y.

(** Pre and Post conditions *)

Definition precondition := prop var.

Definition postcondition := prop pvar.

Definition interp_pre (m0:mem) : precondition -> Prop :=
  interp_prop m0.

Definition interp_pvar (m0:mem) (r:value) (x:pvar) : value :=
   match x with 
   | PV0 x => m0 x
   | PVres => r
   end.

Definition interp_post (m0:mem) (r:value) : postcondition -> Prop :=
  interp_prop (interp_pvar m0 r).

























vcgen/JVMi.v

Set Implicit Arguments.
Require Export Basic.
Require Tree.

(** Bytecode Syntax *)

Definition lbl := positive.

Inductive binstr : Set :=
  | Iload (x:var)
  | Istore (x:var)
  | Ipush (z:Z)
  | Ibop (o:bop) 
  | Iif (c:cop) (l:lbl)
  | Igoto (l:lbl)
  | Ireturn.

Definition bprog := lbl -> binstr.

(** Stack operations *)

Definition stack := list value.

Fixpoint pop (n:nat) (stk:stack) {struct n} : stack :=
   match n, stk with
   | S n, _ :: stk => pop n stk
   | _, _ => stk
   end.

Definition access (n:nat) (stk:stack) : value := 
   nth n stk 0%Z.

(** virtual machine state *)

Record state : Set := State {
    pc:lbl;
    stk:stack;
    m:mem
  }.

(** Bytecode Semantics *)

Section SEMANTICS.

 Variable P : bprog.

 Definition succ : lbl -> lbl := Psucc.
   
 Inductive step : state -> state -> Prop :=
  | Sload : forall x pc stk m,  
      P pc = Iload x ->
      step (State pc stk m) (State (succ pc) (m x :: stk) m)
  | Sstore : forall x pc v stk m,
      P pc = Istore x ->
      step (State pc (v::stk) m) (State (succ pc) stk (update m x v))
  | Spush : forall c pc stk m,
      P pc = Ipush c ->
      step (State pc stk m) (State (succ pc) (c::stk) m)
  | Sbop : forall o v1 v2 pc stk m,
     P pc = Ibop o ->
     step (State pc (v1::v2::stk) m) (State (succ pc) (eval_bop o v1 v2 :: stk) m)
  | Sif_true : forall t k v1 v2 pc stk m,
     P pc = Iif t k ->
     eval_cmp t v1 v2 = true ->
     step (State pc (v1::v2::stk) m) (State (succ pc) stk m)
  | Sif_false : forall t k v1 v2 pc stk m,
     P pc = Iif t k ->
     eval_cmp t v1 v2 = false ->
     step (State pc (v1::v2::stk) m) (State k stk m)
  | Sgoto : forall k pc stk m,
     P pc = Igoto k ->
     step (State pc stk m) (State k stk m).
  (* Remark: No rule for return *)

 Inductive sem : state -> value -> Prop :=
   | Sreturn : forall v pc stk m, 
      P pc = Ireturn ->
      sem (State pc (v::stk) m) v
   | Strans : forall s1 s2 v,
      step s1 s2 ->
      sem s2 v ->
      sem s1 v.

 Definition eval_prog m0 v := sem (State 1%positive nil m0) v.
End SEMANTICS.

(** bytecode assertion *)

Inductive stk_expr : Set :=
  | SEvar
  | SEpush (e:gexpr svar) (os:stk_expr)
  | SEpop (n:nat) (os:stk_expr)

with svar : Set :=
  | SV0 (x:var)
  | SV (x:var)
  | SVaccess (n:nat) (os:stk_expr).

Definition bexpr := gexpr svar.

Definition bassertion := prop svar.

Section Scheme.
  Variable Pe : bexpr -> Prop.
  Variable Pv : svar -> Prop.
  Variable Ps : stk_expr -> Prop.

  Hypothesis H0 : forall x, Pv x -> Pe (Var x).
  Hypothesis H1 : forall c, Pe (Const _ c).
  Hypothesis H2 : forall op e1 e2, Pe e1 -> Pe e2 -> Pe (Op e1 op e2).

  Hypothesis H3 : Ps (SEvar).
  Hypothesis H4 : forall e os, Pe e -> Ps os -> Ps (SEpush e os).
  Hypothesis H5 : forall n os, Ps os -> Ps (SEpop n os).

  Hypothesis H6 : forall x, Pv (SV0 x).
  Hypothesis H7 : forall x, Pv (SV x).
  Hypothesis H8 : forall n os, Ps os -> Pv (SVaccess n os).

  Fixpoint stk_expr_ind2 (os:stk_expr) : Ps os :=
    match os return Ps os with
    | SEvar => H3
    | SEpush e os' =>
      H4  
        (gexpr_ind Pe H1 (fun x => H0 (svar_ind2 x)) 
          (fun e1 He1 o e2 He2 => H2 o He1 He2) e)
        (stk_expr_ind2 os')
    | SEpop n os' =>
      H5 n (stk_expr_ind2 os')
    end 
  with svar_ind2 (x:svar) : Pv x :=
    match x return Pv x with
    | SV0 y => H6 y
    | SV y => H7 y
    | SVaccess n os => H8 n (stk_expr_ind2 os)
    end.

End Scheme.

Section INTERP.

  Variables m0 m : mem.
  Variable stk : stack.

   Fixpoint interp_stack (os:stk_expr) : stack :=
    match os with
    | SEvar => stk
    | SEpush e os => eval_gexpr interp_svar e :: interp_stack os
    | SEpop n os => pop n (interp_stack os)
    end
  with interp_svar (x:svar) : value :=
    match x with
    | SV0 x => m0 x
    | SV x => m x
    | SVaccess n os => nth n (interp_stack os) 0%Z
    end.

 Definition eval_bexpr : bexpr -> value := eval_gexpr interp_svar.

 Definition interp_bassertion : bassertion -> Prop := interp_prop interp_svar.

End INTERP.

(** Normalization operators for stk_expr and svar *)

Fixpoint mkPop (n:nat) (os:stk_expr) {struct n} : stk_expr :=
  match n, os with
  | O, _ => os
  | S n, SEpush _ os => mkPop n os
  | _, _ =>  SEpop n os
  end.

Fixpoint mkAccess (n:nat) (os:stk_expr) {struct n} : bexpr :=
  match os, n with
  | SEpush e os, O => e
  | SEpush _ os, S n => mkAccess n os
  | _, _ => Var (SVaccess n os)
  end.
 
Lemma interp_mkPop : forall m0 m1 stk0 n os,
    interp_stack m0 m1 stk0 (mkPop n os) = interp_stack m0 m1 stk0 (SEpop n os).
Proof.
 intros m0 m1 stk0;induction n;intros os;simpl;trivial.
 destruct os;trivial.
 rewrite IHn;trivial.
Qed.

Lemma interp_mkAccess : forall m0 m1 stk0 n os,
   eval_bexpr m0 m1 stk0 (mkAccess n os) =
   access n (interp_stack m0 m1 stk0 os).
Proof.
 intros m0 m1 stk0;induction n;intros os;simpl;trivial;
 destruct os;trivial.
 rewrite IHn;trivial.
Qed. 

(** Substitution operators *)

Section SUBST_OS.

  Variable new : stk_expr.

  Fixpoint subst_os_stk (os:stk_expr) : stk_expr :=
    match os with
    | SEvar => new
    | SEpush e os => SEpush (map_gexpr subst_os_v e) (subst_os_stk os)
    | SEpop n os => mkPop n (subst_os_stk os)
    end
  with subst_os_v (x:svar) : bexpr :=
    match x with 
    | SV0 x => Var (SV0 x)
    | SV x => Var (SV x)
    | SVaccess n os => mkAccess n (subst_os_stk os)
    end.

End SUBST_OS.

Section SUBST.

  Variable x:var.
  Variable e:bexpr.

  Fixpoint subst_stk (os:stk_expr) : stk_expr :=
    match os with
    | SEvar => SEvar
    | SEpush e os => SEpush (map_gexpr subst_svar e) (subst_stk os)
    | SEpop n os => SEpop n (subst_stk os)
    end
  with subst_svar (y:svar) : bexpr :=
    match y with
    | SV0 y => Var (SV0 y)
    | SV y => if Veq x y then e else Var (SV y)
    | SVaccess n os => Var (SVaccess n (subst_stk os))
    end.

End SUBST.

Definition bsubst_expr (e1:bexpr) (x:var) (e:bexpr) :=
   map_gexpr (subst_svar x e) e1.

Definition bsubst  (p:bassertion) (x:var) (e:bexpr) : bassertion :=
  map_prop (subst_svar x e) p.

Definition bsubst_os_expr (os:stk_expr) (e:bexpr) : bexpr :=
   map_gexpr (subst_os_v os) e.

Definition bsubst_os (p:bassertion) (new:stk_expr) : bassertion:=
   map_prop (subst_os_v new) p.

Lemma interp_svar_bsubst_os :
   forall m0 m stk os x,
   eval_bexpr m0 m stk (subst_os_v os x) =
   interp_svar m0 m (interp_stack m0 m stk os) x.
Proof.
 intros m0 m1 stk0 os x.
 apply (@svar_ind2 
              (fun e => eval_bexpr m0 m1 stk0  (bsubst_os_expr os e) =
                              eval_bexpr m0 m1 (interp_stack m0 m1 stk0 os) e)
              (fun x => eval_bexpr m0 m1 stk0 (subst_os_v os x) =
                              interp_svar m0 m1 (interp_stack m0 m1 stk0 os) x)
              (fun os' => interp_stack m0 m1 stk0 (subst_os_stk os os') =
                                 interp_stack m0 m1 (interp_stack m0 m1 stk0 os) os'));
 simpl;intros;auto.
 rewrite H;rewrite H0;trivial.
 match goal with
 | H: ?X = eval_bexpr _ _ _ _, H0 : ?Y = _ |- ?Z = _ => 
     change Z with (X::Y);rewrite H;rewrite H0;reflexivity end.
 change (subst_os_stk os (SEpop n os0)) with (mkPop n (subst_os_stk os os0)).
 rewrite interp_mkPop.
 match goal with H : ?X = _ |- ?Z = _ => change Z with (pop n X);rewrite H;reflexivity end.
 change (subst_os_v os (SVaccess n os0)) with (mkAccess n (subst_os_stk os os0)).
 rewrite interp_mkAccess;rewrite H;reflexivity.
Qed.

Lemma interp_bexpr_bsubst_os : forall m0 m1 stk0 os e,
   eval_bexpr m0 m1 stk0 (bsubst_os_expr os e) =
   eval_bexpr m0 m1  (interp_stack m0 m1 stk0 os) e.
Proof.
 induction e;simpl;trivial.
 apply interp_svar_bsubst_os.
 rewrite IHe1;rewrite IHe2;trivial.
Qed.

Lemma interp_bsubst_os : forall m0 m stk os Q, 
     interp_bassertion m0 m stk (bsubst_os Q os) =
     interp_bassertion m0 m (interp_stack m0 m stk os) Q.
Proof.
 induction Q;simpl;intros;trivial.
 rewrite IHQ;trivial.
 match goal with
 |  |- interp_cmp c ?X ?Y = _ =>
        change X with (eval_bexpr m0 m1 stk0 (bsubst_os_expr os e1));
        change Y with (eval_bexpr m0 m1 stk0 (bsubst_os_expr os e2));
        repeat rewrite interp_bexpr_bsubst_os;trivial
 end.
 rewrite IHQ1;rewrite IHQ2;tauto.
Qed. 

Lemma eval_subst_update_var :
  forall m0 m1 stk0 x e x0,
    eval_bexpr m0 m1 stk0 (subst_svar x e x0) =
    interp_svar m0 (update m1 x (eval_bexpr m0 m1 stk0 e)) stk0 x0.
Proof.
 intros m0 m1 stk0 x e x0.
 apply (@svar_ind2 
              (fun e1 => eval_bexpr m0 m1 stk0  (bsubst_expr e1 x e) =
                              eval_bexpr m0 (update m1 x (eval_bexpr m0 m1 stk0 e)) stk0 e1)
              (fun x0 => eval_bexpr m0 m1 stk0 (subst_svar x e x0) =
                              interp_svar m0 (update m1 x (eval_bexpr m0 m1 stk0 e)) stk0 x0)
              (fun os => interp_stack m0 m1 stk0 (subst_stk x e os) =
                                interp_stack m0 (update m1 x (eval_bexpr m0 m1 stk0 e)) stk0 os));
   simpl;intros;auto.
 repeat match goal with H: _ = _ |- _ => rewrite H end;trivial.
 match goal with 
    H : ?X = eval_bexpr _ _ _ _ , H0: ?Y = _ |- ?Z = _ => 
      change Z with (X::Y);rewrite H;rewrite H0;trivial end.
 match goal with 
    H : ?X = _ |- ?Y = _ =>
      change Y with (pop n X);rewrite H;trivial end.
 unfold subst_svar, interp_svar, update.
 destruct (Veq x x1);trivial.
 fold (subst_stk x e os).
 match goal with H : ?X = _ |- ?Y = _ =>
     change Y with (access n X);rewrite H;trivial end.
Qed.

Lemma eval_subst_update : 
   forall m0 m1 stk0 x e e1,
     eval_bexpr m0 m1 stk0 (bsubst_expr e1 x e) =
     eval_bexpr m0 (update m1 x (eval_bexpr m0 m1 stk0 e)) stk0 e1.
Proof.
 induction e1;simpl;trivial.
 apply eval_subst_update_var.
 rewrite IHe1_1;rewrite IHe1_2;trivial.
Qed.

Lemma interp_bsubst : forall m0 m stk e x Q,
   interp_bassertion m0 m stk (bsubst Q x e) =
   interp_bassertion m0 (update m x (eval_bexpr m0 m stk e)) stk Q.
Proof.
 induction Q;simpl;trivial.
 rewrite IHQ;trivial.
 assert (X1 := eval_subst_update m0 m1 stk0 x e e1).
 unfold eval_bexpr, bsubst_expr in X1;rewrite X1.
 assert (X2 := eval_subst_update m0 m1 stk0 x e e2).
 unfold eval_bexpr, bsubst_expr in X2;rewrite X2;trivial.
 rewrite IHQ1;rewrite IHQ2;trivial.
Qed.


(* To Move *)
Definition svar_of_pvar (x:pvar) : bexpr:=
  match x with 
  | PVres => Var (SVaccess 0 SEvar) 
  | PV0 x => Var (SV0 x)
  end.

Definition bassertion_of_post : postcondition -> bassertion :=
  map_prop svar_of_pvar.

Lemma eval_svar_pvar : 
  forall m0 m1 v stk0 e,
   eval_gexpr (interp_svar m0 m1 (v :: stk0)) (map_gexpr svar_of_pvar e) =
   eval_gexpr (interp_pvar m0 v) e.
Proof.
 induction e;simpl;trivial.
 destruct x;trivial.
 rewrite IHe1;rewrite IHe2;trivial.
Qed.

Lemma interp_post_bassertion : forall m0 m v stk Post0,
  interp_bassertion m0 m (v :: stk) (bassertion_of_post Post0) =
  interp_post m0 v Post0.
Proof.
 induction Post0;simpl;auto.
 rewrite IHPost0;trivial.
 repeat rewrite eval_svar_pvar;trivial.
 rewrite IHPost0_1;rewrite IHPost0_2;trivial.
Qed.

Lemma eval_svar_var0 : forall m0 e,
    eval_gexpr (interp_svar m0 m0 nil)
     (map_gexpr (fun x : var => Var (SV0 x)) e) =
   eval_gexpr m0 e.
Proof.
 induction e;simpl;trivial.
 rewrite IHe1;rewrite IHe2;auto.
Qed.

Definition bassertion_of_pre (pre:precondition) : bassertion :=
  map_prop (fun x => Var (SV0 x)) pre.

Lemma interp_pre_bassertion : forall Pre0 m0, 
     interp_pre m0 Pre0 =
     interp_bassertion m0 m0 nil (bassertion_of_pre Pre0).
Proof.
 induction Pre0;simpl;intros;trivial.
 rewrite IHPre0;trivial.
 repeat rewrite eval_svar_var0;trivial.
 rewrite IHPre0_1;rewrite IHPre0_2;trivial.
Qed.

(** Verification conditions generator *)

Section WP.

  Variable P : bprog.

  Variable Post : postcondition.

  Section WP_i.

  Variable wp_l : lbl -> bassertion.

  Definition wp_i (pc:lbl) : bassertion :=
    match P pc with
    | Iload x => 
      let post := wp_l (succ pc) in
      bsubst_os post (SEpush (Var (SV x)) SEvar)
    | Istore x =>
      let post := wp_l (succ pc) in
      bsubst (bsubst_os post (SEpop 1 SEvar)) x (Var (SVaccess O SEvar))
    | Ipush z =>
      let post := wp_l (succ pc) in
      bsubst_os post (SEpush (Const _ z) SEvar)
    | Ibop op => 
      let post := wp_l (succ pc) in
      bsubst_os post (SEpush (Op (Var (SVaccess 0 SEvar)) op (Var (SVaccess 1 SEvar))) (SEpop 2 SEvar))
    | Iif c k =>   
      let post1 := bsubst_os (wp_l (succ pc)) (SEpop 2 SEvar) in
      let post2 := bsubst_os (wp_l k) (SEpop 2 SEvar) in
      let cond := Lcmp (Var (SVaccess 0 SEvar)) c (Var (SVaccess 1 SEvar)) in
      Lop (Lop cond Limp post1) Land  (Lop (Lnot cond) Limp post2)
    | Igoto k => wp_l k
    | Ireturn => bassertion_of_post Post
    end.

 End WP_i.

 Variable annot : lbl -> option bassertion.

 Fixpoint wp_l (n:nat) (pc:lbl) {struct n} : bassertion :=
    match annot pc with
     | Some assert => assert
     | None =>
       match n with
       | 0 => Lfalse _
       | S n => wp_i (wp_l n) pc
     end
   end.

End WP.

Definition interp_cond (C:bassertion) :=
  forall m0 m stk, interp_bassertion m0 m stk C.

Fixpoint  interp_lcond (P:Prop) (l : list bassertion) :=
  match l with
  | nil => P
  | C::l => interp_lcond (P /\ interp_cond C) l
  end.

Definition interp_vc (vc:bassertion * list bassertion) :=
  let (init, phi) := vc in
  interp_lcond (forall m0, interp_bassertion m0 m0 nil init) phi. 

Section CORRECTION.

 Definition annot_tbl := Tree.tree bassertion.

 Variable P : bprog.
 Variable Pre : precondition.
 Variable tbl : annot_tbl.
 Variable Post : postcondition.
 Variable gaz : nat.

 Definition annot pc :=  Tree.get _ pc tbl.
 Definition wp_lbl := wp_l P Post annot gaz.
 Definition wp_instr := wp_i P Post wp_lbl.
 

 Definition init_cond := 
   let pre := wp_lbl 1%positive in
   Lop (bassertion_of_pre Pre) Limp pre.
 
 Definition bvcgen :=
   let cond := Tree.fold _ _ (fun pc assert res => 
       (Lop assert Limp (wp_instr pc))::res) tbl nil in
   (init_cond, cond).

 Definition valid_prog := interp_vc bvcgen.

 Hypothesis vp : valid_prog.

 Lemma interp_lcond_correct : 
   forall phi res,
     interp_lcond res phi ->
     res /\
     forall Q, In Q phi -> interp_cond Q.
 Proof.
  induction phi;simpl;intros.
  split;auto. intros Q H0;elim H0.
  destruct (IHphi _ H) as ((H1, H2), H3);clear IHphi H;intuition.
  subst;auto.
 Qed.

 Lemma wp_lbl_wp_instr_aux : 
   forall tbl0 res prefix,
      forall Q, 
       In Q res \/
       (exists pc, exists assert, 
         Tree.get _ pc tbl0 = Some assert /\
         Q = Lop assert Limp (wp_instr (prefix pc))) ->
       In Q 
         (Tree.fold_aux _ _ 
            (fun pc assert res => 
               (Lop assert Limp (wp_instr pc))::res) prefix tbl0 res).
 Proof.
  induction tbl0;simpl;intros;auto.
  apply IHtbl0_1.
  destruct H as [H | (pc1, (Q1, (H1, H2)))] .
  left;apply IHtbl0_2.
  left;destruct o;simpl;auto.
  subst;destruct pc1.
  right;exists pc1;exists Q1;auto.  
  left;apply IHtbl0_2.
  right;exists pc1;exists Q1;auto.
  left;subst;apply IHtbl0_2;simpl;auto.
  destruct H as [H | (pc1, (Q1, (H1, H2)))];auto.
  destruct pc1;discriminate H1.
 Qed. 
     
 Lemma valid_init : forall m0,  interp_bassertion m0 m0 nil init_cond.
 Proof.
  unfold valid_prog, bvcgen in vp;simpl in vp.
  destruct (interp_lcond_correct _ _ vp).
  unfold init_cond;simpl;intros.
  apply H;auto.
 Qed.

 Lemma wp_lbl_wp_instr_annot : forall pc m0 m stk Q,
   annot pc = Some Q -> 
   interp_bassertion m0 m stk Q -> interp_bassertion m0 m stk (wp_instr pc).
 Proof.
  intros pc0 m0 m1 stk0 Q H.
  assert (In (Lop Q Limp (wp_instr pc0)) nil \/ (exists pc, exists assert, 
         Tree.get _ pc tbl = Some assert /\
         Lop Q Limp (wp_instr pc0) = Lop assert Limp (wp_instr pc))).
    right;exists pc0;exists Q;auto.
  assert (X:= @wp_lbl_wp_instr_aux tbl nil (fun x => x) _ H0).
  destruct (interp_lcond_correct _ _ vp).
  exact (H2 _ X m0 m1 stk0).
 Qed.

 Lemma wp_lbl_wp_instr: forall pc m0 m stk n,
   interp_bassertion m0 m stk (wp_l P Post annot n pc) ->
   exists n', interp_bassertion m0 m stk (wp_i P Post (wp_l P Post annot n') pc).
 Proof.
  intros pc0 m0 m1 stk0 n H.
  destruct n;simpl in H;(case_eq (annot pc0); [intros Q Heq | intros Heq];
   rewrite Heq in H).
  exists gaz;exact (wp_lbl_wp_instr_annot _ _ _ _ Heq H).
  elim H.
  exists gaz;exact (wp_lbl_wp_instr_annot _ _ _ _ Heq H).
  exists n;trivial.
 Qed.

 Lemma wp_instr_step : forall m0 s1 s2 n,
   step P s1 s2 ->
   interp_bassertion m0 (m s1) (stk s1) (wp_i P Post (wp_l P Post annot n) (pc s1)) ->
   interp_bassertion m0 (m s2) (stk s2) (wp_l P Post annot n (pc s2)).
 Proof.
  intros m0 s1 s2 n Hstep;inversion Hstep;subst;clear Hstep;intros Hwp;
  unfold wp_i in Hwp;simpl in Hwp;
  match goal with H:P ?pc = _ |- _ => try rewrite H in Hwp end;simpl in Hwp |- *.
  rewrite interp_bsubst_os in Hwp;trivial.
  rewrite interp_bsubst in Hwp;rewrite interp_bsubst_os in Hwp;trivial.
  rewrite interp_bsubst_os in Hwp;trivial.
  rewrite interp_bsubst_os in Hwp;trivial.
  destruct Hwp as (Hwp, _).
  assert (HH := Hwp (eval_cmp_true_interp _ _ _ H0));clear Hwp.
  rewrite interp_bsubst_os in HH;trivial.
  destruct Hwp as (_, Hwp).
  assert (HH := Hwp (eval_cmp_false_interp _ _ _ H0));clear Hwp.
  rewrite interp_bsubst_os in HH;trivial.
  trivial.
 Qed.

 Lemma wp_instr_trans : forall (m0 : mem) (s : state) (v : value),
    sem P s v ->
     forall n, 
       interp_bassertion m0 (m s) (stk s)
         (wp_i P Post (wp_l P Post annot n) (pc s)) -> 
       interp_post m0 v Post.
 Proof.
  induction 1.
  unfold wp_i;simpl;rewrite H.
  rewrite interp_post_bassertion;trivial.
  intros n H1.
  assert (X:= wp_instr_step _ _ H H1).
  destruct (wp_lbl_wp_instr _ _ _ _ _ X) as (n', H2).
  apply (IHsem _ H2).
 Qed.
   
 Lemma vcgen_correct : forall m0,
    interp_pre m0 Pre ->
    forall v,
       eval_prog P m0 v ->
       interp_post m0 v Post.
 Proof.
  intros m0 Hpre v Hsem.
  rewrite interp_pre_bassertion in Hpre.
  assert (X:= valid_init m0 Hpre).
  destruct (wp_lbl_wp_instr _ _ _ _ _ X) as (n, H).
  apply wp_instr_trans with (State 1%positive nil m0) n;trivial.
 Qed.

End CORRECTION.

Check vcgen_correct.

Definition cons := Tree.set binstr.
Open Scope positive_scope.

Notation n := 1.
Notation i := 2.

Definition Bsum :=
let code :=
cons 1 (Ipush 0)          (
cons 2 (Istore i)            (
cons 3 (Iload n)            (
cons 4 (Iload i)             (
cons 5 (Iif Olt 11)         (
cons 6 (Ipush 1)          (
cons 7 (Iload i)             (
cons 8 (Ibop Oadd)      (
cons 9 (Istore i)            (
cons 10 (Igoto 3)          (
cons 11 (Iload i)            (
cons 12 Ireturn     (Tree.Empty _)        ))))))))))) in
fun k => 
   match Tree.get _ k code with
   | Some ii => ii
   | _ => Ireturn 
   end.

Definition sum_pre := (Lcmp (Const _ 0) Ole (Var n)).

Definition sum_annot :=
   Tree.set _ 3
      (Lop 
          (Lcmp (Var (SV n)) Oeq (Var (SV0 n)))
        Land
          (Lcmp (Var (SV i)) Ole (Var (SV0 n)))) (Tree.Empty _).

Definition sum_post :=  (Lcmp  (Var PVres) Oeq (Var (PV0 n))).

Ltac set_vc := 
  match goal with
  |- interp_vc ?X => 
      let vc := fresh "vc" in set (vc := X)
   end.

Ltac red_interp_vc :=
 cbv beta iota zeta delta 
    [interp_vc interp_lcond interp_cond 
      interp_svar interp_bassertion interp_prop eval_gexpr
      interp_lop eval_bop interp_cmp].

Lemma sum_spec : 
     forall (m0:mem),  (0 <= m0 n)%Z ->  
      forall v, eval_prog Bsum m0 v ->
         v = m0 n.
Proof.
intros.
Check vcgen_correct.
refine (@vcgen_correct Bsum sum_pre sum_annot sum_post 100%nat
               _ m0 H v H0).
unfold valid_prog.

(* vm_compute. (* bad idea *) *)

set (vc := bvcgen Bsum sum_pre sum_annot sum_post 100). 
vm_compute in vc.
unfold vc;clear vc.

red_interp_vc.



auto with zarith.

Qed.

Print sum_spec.

vcgen/Makefile

##############################################################################
##                 The Calculus of Inductive Constructions                  ##
##                                                                          ##
##                                Projet Coq                                ##
##                                                                          ##
##                     INRIA                        ENS-CNRS                ##
##              Rocquencourt                        Lyon                    ##
##                                                                          ##
##                                  Coq V7                                  ##
##                                                                          ##
##                                                                          ##
##############################################################################

# WARNING
#
# This Makefile has been automagically generated by coq_makefile
# Edit at your own risks !
#
# END OF WARNING

#
# This Makefile was generated by the command line :
# coq_makefile -f Make -o Makefile 
#

##########################
#                        #
# Variables definitions. #
#                        #
##########################

CAMLP4LIB=`camlp4 -where`
COQSRC=-I $(COQTOP)/kernel -I $(COQTOP)/lib \
  -I $(COQTOP)/library -I $(COQTOP)/parsing \
  -I $(COQTOP)/pretyping -I $(COQTOP)/interp \
  -I $(COQTOP)/proofs -I $(COQTOP)/syntax -I $(COQTOP)/tactics \
  -I $(COQTOP)/toplevel -I $(COQTOP)/contrib/correctness \
  -I $(COQTOP)/contrib/extraction -I $(COQTOP)/contrib/field \
  -I $(COQTOP)/contrib/fourier -I $(COQTOP)/contrib/graphs \
  -I $(COQTOP)/contrib/interface -I $(COQTOP)/contrib/jprover \
  -I $(COQTOP)/contrib/omega -I $(COQTOP)/contrib/romega \
  -I $(COQTOP)/contrib/ring -I $(COQTOP)/contrib/xml \
  -I $(CAMLP4LIB)
ZFLAGS=$(OCAMLLIBS) $(COQSRC)
OPT=
COQFLAGS=-q $(OPT) $(COQLIBS) $(OTHERFLAGS) $(COQ_XML)
COQC=$(COQBIN)coqc
GALLINA=gallina
COQDOC=$(COQBIN)coqdoc
CAMLC=ocamlc -c
CAMLOPTC=ocamlopt -c
CAMLLINK=ocamlc
CAMLOPTLINK=ocamlopt
COQDEP=$(COQBIN)coqdep -c
GRAMMARS=grammar.cma
CAMLP4EXTEND=pa_extend.cmo pa_ifdef.cmo q_MLast.cmo
PP=-pp "camlp4o -I . -I $(COQTOP)/parsing $(CAMLP4EXTEND) $(GRAMMARS) -impl"

#########################
#                       #
# Libraries definition. #
#                       #
#########################

OCAMLLIBS=-I .
COQLIBS=-I .

###################################
#                                 #
# Definition of the "all" target. #
#                                 #
###################################

VFILES=Tree.v\
  Basic.v\
  While.v\
  JVMi.v\
  Prime.v
VOFILES=$(VFILES:.v=.vo)
VIFILES=$(VFILES:.v=.vi)
GFILES=$(VFILES:.v=.g)
HTMLFILES=$(VFILES:.v=.html)
GHTMLFILES=$(VFILES:.v=.g.html)

all: Tree.vo\
  Basic.vo\
  While.vo\
  JVMi.vo\
  Prime.vo

spec: $(VIFILES)

gallina: $(GFILES)

html: $(HTMLFILES)

gallinahtml: $(GHTMLFILES)

all.ps: $(VFILES)
	$(COQDOC) -ps -o $@ `$(COQDEP) -sort -suffix .v $(VFILES)`

all-gal.ps: $(VFILES)
	$(COQDOC) -ps -g -o $@ `$(COQDEP) -sort -suffix .v $(VFILES)`



####################
#                  #
# Special targets. #
#                  #
####################

.PHONY: all opt byte archclean clean install depend html

.SUFFIXES: .v .vo .vi .g .html .tex .g.tex .g.html

.v.vo:
	$(COQC) $(COQDEBUG) $(COQFLAGS) $*

.v.vi:
	$(COQC) -i $(COQDEBUG) $(COQFLAGS) $*

.v.g:
	$(GALLINA) $<

.v.tex:
	$(COQDOC) -latex $< -o $@

.v.html:
	$(COQDOC) -html $< -o $@

.v.g.tex:
	$(COQDOC) -latex -g $< -o $@

.v.g.html:
	$(COQDOC) -html -g $< -o $@

byte:
	$(MAKE) all "OPT="

opt:
	$(MAKE) all "OPT=-opt"

include .depend

.depend depend:
	rm -f .depend
	$(COQDEP) -i $(COQLIBS) $(VFILES) *.ml *.mli >.depend
	$(COQDEP) $(COQLIBS) -suffix .html $(VFILES) >>.depend

install:
	mkdir -p `$(COQC) -where`/user-contrib
	cp -f $(VOFILES) `$(COQC) -where`/user-contrib

Makefile: Make
	mv -f Makefile Makefile.bak
	$(COQBIN)coq_makefile -f Make -o Makefile


clean:
	rm -f *.cmo *.cmi *.cmx *.o $(VOFILES) $(VIFILES) $(GFILES) *~
	rm -f all.ps all-gal.ps $(HTMLFILES) $(GHTMLFILES)

archclean:
	rm -f *.cmx *.o

html:

# WARNING
#
# This Makefile has been automagically generated by coq_makefile
# Edit at your own risks !
#
# END OF WARNING

Set Printing Implicit.
Require Import ZArith.
Open Scope Z_scope.


(***** A proof of 2 + 2 = 4 ******)

Lemma waou : 2+2 = 4.
simpl.
apply refl_equal. (* reflexivity.*)
Qed.


Print waou.
Check refl_equal.
Check (@refl_equal Z 4).



(***** A simple primality test ********)

Require Import Znumtheory.

Print prime.
Print rel_prime.

(* prime p := forall n, 1 <= n < p -> Zis_gcd n p 1 *) 

(* Given n p testing that "Zis_gcd n p 1" is trivial *)

Definition test_gcd p q:= Zeq_bool (Zgcd p q) 1.

Eval compute in test_gcd 2 7.
Eval compute in test_gcd 4 6.

(* Checking the quantification :
      forall n, 1 <= n < p,  test_gcd p n = true
*)

Fixpoint Forall (test:Z -> bool) (x:Z) (p:positive) {struct p} : bool :=
  match p with
  | xH => test x
  | xO p' => 
    if Forall test x p' then Forall test (x + Zpos p') p'
    else false
  | xI p' =>
    if Forall test x p' then 
     if Forall test (x + Zpos p') p' then test (x + Zpos (xO p'))
     else false
    else false
  end.

Lemma Forall_correct : forall test p x, 
     Forall test x p = true -> 
     forall z, x <= z < x + Zpos p -> test z = true.
Proof.
  induction p;simpl Forall;intros x;
  try rewrite Zpos_xI;try rewrite Zpos_xO.
  case_eq (Forall test x p);intros Heq1;try (intros;discriminate).
  case_eq (Forall test (x+Zpos p) p);try (intros;discriminate);
   intros Heq2 Heq3 z H.
  assert 
    (   x <= z < x + Zpos p 
     \/ x + Zpos p <= z < x + 2 * Zpos p
     \/ z = x + 2 * Zpos p)
    by omega.
  destruct H0 as [ H1 | [ H2 | H3 ] ].
  apply IHp with x;trivial.
  apply IHp with (x + Zpos p);trivial. 
    ring_simplify (x + Zpos p + Zpos p). trivial.
  rewrite H3;trivial.
  case_eq (Forall test x p);intros Heq1;intros;try discriminate.
  assert (x <= z < x + Zpos p \/ x + Zpos p <= z < x + 2 * Zpos p)
    by omega.
  destruct H1 as [ H1 | H2 ].
  apply IHp with x;trivial.
  apply IHp with (x + Zpos p);trivial. 
    ring_simplify (x + Zpos p + Zpos p). trivial.
  intros;assert (z = x) by omega.
  rewrite H1;trivial.
Qed.

(* Our semi-decision algorithm, in fact a decision algorithm *)
Definition test_prime p := 
  if Zlt_bool 1 (Zpos p) then
     Forall (test_gcd (Zpos p)) 1 (p - 1)
  else false.

(* Test our primality test *)

Eval compute in test_prime 2.
Eval compute in test_prime 3.
Eval compute in test_prime 4.
Eval compute in test_prime 5.

(* Correctness proof *)

Lemma test_prime_correct : 
   forall p, test_prime p = true -> prime (Zpos p).
Proof.
  intros p;unfold test_prime.
  case_eq (Zlt_bool 1 (Zpos p));intros Heq;try (intros;discriminate).
  intros.
  assert (X: 1 < Zpos p).
   unfold Zlt,Zlt_bool in *.
   destruct (1 ?= Zpos p);trivial;try discriminate Heq.
  constructor;trivial.
  intros;unfold rel_prime.
  apply Zis_gcd_sym.  
  assert (Zgcd (Zpos p) n = 1).
   apply Zeqb_ok.
   unfold test_prime, test_gcd in H0.
   apply (Forall_correct _ _ _ H n).
   rewrite Pminus_Zminus.
    omega.
   destruct p;trivial.
  rewrite <- H1;apply Zgcd_is_gcd.
Qed.

(* The year of the french revolution is a prime number !!! *)
Lemma prime1789 : prime 1789.
Proof.
 apply test_prime_correct.
 compute. 
 trivial.
Time Qed.
Print prime1789.

Lemma prime1789_vm : prime 1789.
Proof.
 apply test_prime_correct.
 vm_compute. 
 trivial.
Time Qed.

(* How look the proof term *)
Print prime1789.
Print prime1789_vm.

Lemma prime48611 : prime 48611.
Proof.
  apply test_prime_correct.
  vm_compute. trivial.
Time Qed.

Print prime48611.
(* It is almost the same proof *)

Require Export ZArith.

Section Tree.
  
  Variable (A:Set).

  Inductive tree : Set :=
    | Node : option A -> tree -> tree -> tree
    | Empty : tree.

  Fixpoint get (k:positive) (t:tree) {struct t} : option A :=
   match k, t with
   | xH, Node o _ _ => o
   | xI k, Node _ l _ => get k l
   | xO k, Node _ _ r => get k r
   | _, _ => None
   end.

  Fixpoint set (k:positive) (a:A) (t:tree) {struct k} : tree :=
   match k, t with
   | xH, Empty => Node (Some a) Empty Empty
   | xH, Node _ l r => Node (Some a) l r
   | xI k, Empty => Node None (set k a Empty) Empty
   | xI k, Node o l r => Node o (set k a l) r
   | xO k, Empty => Node None Empty (set k a Empty)
   | xO k, Node o l r => Node o l (set k a r)
   end.
 
 Section FOLD.
   Variable R:Set.
   Variable f : positive -> A -> R -> R.

   Fixpoint fold_aux (prefix:positive -> positive) (t:tree) (res:R) : R :=
    match t with
    | Empty => res
    | Node o l r =>
      let res := match o with None => res | Some a => f (prefix xH) a res end in
      fold_aux (fun p => prefix (xI p)) l
         (fold_aux (fun p => prefix (xO p)) r res)
    end.   

  Definition fold := fold_aux (fun p => p).

 End FOLD.
               
End Tree.

Set Implicit Arguments.

Require Export Basic.


(**************************)
(** Source Language          **)
(**************************)

Definition expr := gexpr var.

Inductive cmp : Set :=
 | Cmp (e1:expr) (c:cop) (e1:expr).

Inductive lvar : Set :=
  | LV0 (x:var)
  | LV (x:var).

Definition assertion := prop lvar.

Inductive instr : Set :=
  | Skip
  | Assign (x:var) (e:expr)
  | Seq (i1:instr) (i2:instr)
  | If (t:cmp) (i1 i2:instr)
  | While (t:cmp) (inv:assertion) (body:instr).

Definition prog := (instr * expr)%type.

Definition lvar_eq x y :=
  match x, y with
  | LV0 x, LV0 y => Veq x y  
  | LV x, LV y => Veq x y 
  | _, _ => false
  end.


Definition aexpr_of_pexpr :=  map_gexpr (fun x => Var (LV x)).

Definition asubst p x e :=
  let ae := aexpr_of_pexpr e in
  map_prop (fun y => if lvar_eq x y then ae else Var  y) p.

Definition subst p x := asubst p (LV x).

Definition subst_res p e :=
  let ae := aexpr_of_pexpr e in
  map_prop (fun y => match y with PV0 x => Var (LV0 x) | PVres => ae end) p.
 
Definition mkCond (t: cmp) : assertion :=
  let (e1, c, e2) := t in
  Lcmp (aexpr_of_pexpr e1) c (aexpr_of_pexpr e2).

(* Verification condition generator *)

Fixpoint wp (i:instr) (post:assertion) {struct i} : assertion * list assertion :=
  match i with 
  | Skip => (post, nil)
  | Assign x e => (subst post x e, nil)
  | Seq i1 i2 =>
     let (pre2, phi2) := wp i2 post in
     let (pre1, phi1) := wp i1 pre2 in
     (pre1, phi1 ++ phi2)
  | If t i1 i2 => 
    let (pre1, phi1) := wp i1 post in
    let (pre2, phi2) := wp i2 post in
    let cond := mkCond t in
    (Lop     (Lop cond Limp pre1)
       Land (Lop (Lnot cond) Limp pre2),
     phi1 ++ phi2)
  | While t inv i =>
     let (pre, phi) := wp i inv in
     let cond := mkCond t in
     (inv,
      (Lop
          inv Limp
          (Lop 
             (Lop cond Limp pre)
             Land 
             (Lop (Lnot cond) Limp post))) :: phi)
   end.
(*
Definition pre_of_assert (a:assertion) : precondition :=
  map_prop (fun x => match x with Ainit x => Var x | Acurrent x => Var x end) a.
*) 

Definition assert_of_pre (pre:precondition) : assertion :=
  map_prop (fun x => Var (LV0 x)) pre.

Definition vcgen (p:prog) (Pre:precondition) (Post:postcondition) : assertion * list assertion :=
  let (i, e) := p in
  let (pre, phi) := wp i (subst_res Post e) in
  (Lop (assert_of_pre Pre) Limp pre,  phi).

Open Scope positive_scope.
Notation n := (Var 1).
Notation i := (Var 2).


Definition Ssum : prog :=
  (Seq (Assign (* i *) 2 (Const _ 0))
           (While (Cmp i Ole n)
                     (Lcmp (Var (LV 2)) Ole (Var (LV0 1))) 
                     (Assign (* i *) 2 (Op i Oadd (Const _ 1)))),
    i).

Eval vm_compute in vcgen Ssum (Lcmp (Const _ 0) Ole n)  (Lcmp (Var (PV0 1)) Oeq (Var PVres)).
                          
           



 





